
Sudoku Sat - Pràctica Prolog

Tutor de la pràctica : Mateu Villaret

Francesc Xavier Bullich Parra i Marc Sànchez Pifarré,
GEINF (UDG-EPS)

2018-06-13

Contents

Informe sobre la pràctica Prolog.	3
Propietats Sudoku	3
Predicats de la pràctica.	4
Resol	4
Tauler Sudoku	5
Fes Tauler	6
Genera llista	7
Separa Per N	8
Treu N	9
Inicialitzar	9
Codifica sudoku	10
K dominis	11
Exactament Un	12
Negat	13
Monta Parelles	13
Combina	14
All Diff Files	15
All Diff	15
Mat Transposa	17
Transposar Fila	18
Mat Unió	18
All Diff Columnes	19
All Diff Quadrats	20
Subset	21
Extreu	23
Dins Rang	25
Sat	26
Decideix	28
Simplif	29
Mostra Sudoku	30
Pinta Separador	31
Pinta Casella	32

Exemples i jocs de proves.	32
SUDOKUS AMB SOLUCIÓ	33
sudoku1	33
sudoku2	34
SUDOKUS SENSE SOLUCIÓ	35
sudokuNoQ	35
sudokuNoF	36
sudokuNoC	37
Comentaris sobre la solució i aspectes positius.	38
Abstracció	38
Talls.	38
L'us de l'append	38
Cerques	38
Detectar clàusula unitària	39
Construir i desconstruir Llistes	39
Conclusió	39

Informe sobre la pràctica Prolog.

En aquest informe s'hi allotja tot el codi degudament documentat sobre la pràctica prolog anomenada com sudokusat. Aquest informe es construeix sobre un exemple en concret de un sudoku de $n=4$. Amb aquest exemple s'anirà escenificant què ocorre a cada un dels predicats que considerem més important.

També consta d'un apartat d'exemples d'execució amb sudokus $N=9$.

Propietats Sudoku

Com a propietats importants d'un sudoku tenim :

- Només podem fer sudokus de tamany N on N és un nombre que té una arrel entera. Per tant es poden fer sudokus de $N=4$, $N=9$, $N=16$...

Predicats de la pràctica.

En aquest apartat es mostren tots els predicats que no son el mostra, el mostraM i el mostraLinia. S'expliquen i es posa un exemple d'execució de cada un d'ells. Així es pot comprovar el seu funcionament i evaluar si el tall està ben posat.

Resol

Aquest predicat seria com el main del programa, és qui delega feina als predicats de manera ordenada i qui estructura les execucions per que el sat rebi els paràmetres que requereix i per que els sudokus es pintin de manera correcte.

Predicat

S'hi ha afegit en tot el codi una sèrie de writes per modificar la visualització del sudoku i fer-la més visual.

```

1  %%%%%%%%%%%%%%
2  % resol(N,Inputs)
3  % Donada la N del sudoku (NxN), i una llista d'inputs,
4  % -> es mostra la solucio per pantalla si en te o es diu que no en te.
5  resol(N,Inputs):- nl, write('NEM A RESOLDRE EL SUDOKU : '),nl,
6                      write('.....'),nl,
7                      mostraSudoku(N,Inputs), taulerSudoku(N, Inputs, T, C0
8                      ), codificaSudoku(N,T,C0,CNF),
                      sat(CNF,[],M), mostra(M,N).
```

Execució

Veiem l'execució següent que és la que utilitzarem a tot l'informe per explicar de manera adequada el funcionament de cada un dels predicats.

```

1  | ?- resol(4,[c(1,1,3),c(2,2,1),c(4,4,3)]).
2
3  NEM A RESOLDRE EL SUDOKU :
4  .....
5  -----
6  |3| | | |
7  -----
8  | |1| | |
9  -----
10 | | | | |
11 -----
```

```

12 | | | |3|
13 -----
14
15 SAT!!, SUDOKU PISCINAS!!!
16 .....
17 -----
18 |3|2|1|4|
19 -----
20 |4|1|3|2|
21 -----
22 |2|3|4|1|
23 -----
24 |1|4|2|3|
25 -----

```

Crides internes

- Tauler Sudoku
- Codifica Sudoku
- Sat
- Mostra

Tauler Sudoku

És un predicat que s'encarrega de generar un tauler a partir d'un nombre i una sèrie de inputs. Aquest tauler està representat en forma de llista de llistes de K-domins. On cada K-Domini es representa com a 4 variables que poden prendre el valor cert o fals. Llavors per representar que un valor és cert deixarem la variable com a enter positiva i per representar que és falça com un enter negatiu.

Predicat :

```

1 %%%%%%%%%%%%%%% DONE MACARRONE!!
2 % taulerSudoku(N,C,Tauler,CNF)
3 % Donat el domini de les caselles (N), una inicialitzacio del Sudoku (
  de forma [c[1,2,5],...]),
4 % -> Tauler serà Llista de llistes de N-dominis del tauler,
5 % -> i CNF codifica els valors assignats amb clausules unitaries que
  forcen valor a les caselles inicialitzades
6 taulerSudoku(N,C,Tauler,CNF):-
7   % Genera totes les possibles variables per cada domini dins de cada
  casella i cada llista.
8   festauler(N,Tauler),

```

```

9    % Agafa les caselles passades com a inputs C i genera la CNF que
      codifica aquestes caselles com a fixes.
10   inicialitzar(N,C,CNF).

```

Execució

```

1  ?- taulerSudoku(4, [c(1,1,3),c(2,2,1),c(4,4,3)], T, CNF).
2  T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
      [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
      31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
      [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
      [61|...]]],
3  CNF = [[3], [21], [63]].

```

En el resultat de l'execució podem veure que T té una llista de files on cada fila té una llista de caselles i on finalment per cada casella hi ha codificat la possibilitat de que prengui el valor 1, 2, 3 o 4 en forma de variables. Per la casella Fila 1 columna 1 el fet que hi hagi un 1 es representa amb un 1. A la casella Fila 1 columna 2 el fet que hi hagi un 1 es representa amb la variable 5.

D'aquesta manera CNF serà el llistat de clàusules que codifiquen que hem forçat un 3 a la fila 1, columna 1, un 1 a la fila 2 columna 2 i un 3 a la fila 4 columna 4.

Crides internes

Es fa la crida al festauler i al inicialitzar.

- Fes Tauler
- Inicialitzar

Fes Tauler

La idea és simple, es genera una llista de K on K és N^3 nombres. Un cop generada la llista la dividim en grups de N, dos vegades. És a dir sobre una llista de K varibales generem una llista LLN de K/N Llistes de N variables cada llista. I sobre la llista de K/N llistes dividim altre cop per N llistes de llistes quedant un llista de N llistes que té N llistes de N dominis on tot plegat son K dominis.

Predicat

```

1  %%%%%%%%%%%%%% DONE MACARRONE!!
2  % festauler(N,T)
3  % Donat el domini de les caselles (N),
4  % -> T sera una llista de N llistes de N llistes de N nombres
5  %     es a dir, una llista de files de N-dominis.

```

```

6 %   El primer N-domini tindra la forma [1,2,...N] i així
    successivament de fila en fila fins
7 %   a l'últim N-domini corresponent a la casella (N,N) que sera [N*N*(
    N-1)+1,N*N*(N-1)+2,...N*N*N]
8 % festauler(0, []). %F is N*N, K is F*N
9 festauler(N, T):- K is N*N*N, generaLlista(1, K, LL), separaPerN(N, LL,
    LLN), separaPerN(N, LLN, T), !.

```

Execució

```

1 ?- festauler(4,T).
2 T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
    [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
    31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
    [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
    [61|...]]].

```

Crides internes

- Genera Llista
- Separa Per N

Genera llista

Generar una llista de K variables on K és N^3 .

```

1 %%%%%%%%%%%%%% DONE MACARRONE!!
2 % generaLlista(P, U, LL)
3 % Genera una llista de números de P fins a U i ho posa a LL
4 % -> P < U.
5 % -> P és l'inici
6 % -> U és el nombre final
7 % -> LL és la llista d'enters de P fins a U. [P..U]
8 generaLlista(P, P, [P]).
9 generaLlista(P, U, LL):- P < U, K is P+1, generaLlista(K, U, R),
    append([P], R, LL),!.

```

Execució

```

1 ?- generaLlista(1, 64, LL).
2 LL = [1, 2, 3, 4, 5, 6, 7, 8, 9|...].

```


Separa Per N

Donats un N i una llista de K elements, divideix la llista de K elements en X subllistes de N elements.

Predicat

```

1 %%%%%%%%%%%%%% DONE MACARRONE!!
2 % separaPerN(N, L, R)
3 % Donada una llista L d'elements, R serà L dividida en K parts de N
   elements cada part.
4 % -> N és el nombre d'elements que han de tenir les subllistes
5 % -> L és la llista a tractar
6 % -> R és una llista de llistes de N elements cada subllista.
7 separaPerN(_, [], []).
8 separaPerN(N, L, R):- treuN(N, L, EXTRETS, CUA), separaPerN(N, CUA,
   NOUEXTRETS), append([EXTRETS], NOUEXTRETS, R), !.
```

Execució

```

1 ?- generaLlista(1, 64, LL), separaPerN(4, LL, RESULTAT).
2 LL = [1, 2, 3, 4, 5, 6, 7, 8, 9|...],
3 RESULTAT = [[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15,
   16], [17, 18, 19, 20], [21, 22, 23|...], [25, 26|...], [29|...],
   [...|...]|...].
```

Un cop feta aquesta crida està tot dividit en caselles, s'ha de tornar a fer la mateixa crida per tenir dividida la llista dos vegades i així obtenir les files a partir de les caselles.

```

1 ?- generaLlista(1, 64, LL), separaPerN(4, LL, LL2), separaPerN(4, LL2,
   RESULTAT).
2 LL = [1, 2, 3, 4, 5, 6, 7, 8, 9|...],
3 LL2 = [[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16],
   [17, 18, 19, 20], [21, 22, 23|...], [25, 26|...], [29|...],
   [...|...]|...],
4 RESULTAT = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15,
   16]], [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29,
   30, 31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
   [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
   [61|...]]].
```

Crides internes

- Treu N

Treu N

Divideix una llista en EXTRETS i CUA on EXTRETS és una llista de N elements de L i CUA és la resta expressada com a llista.

Predicat

```

1 %%%%%%%%%%%%%%% DONE MACARRONE!!
2 % treuN(N, L, EXTRETS, CUA)
3 % Divideix L en EXTRETS i CUA, on EXTRETS conté N elements de L i CUA
  la resta.
4 % N -> nombre d'elements a extreure
5 % L -> Llista d'elements a tractar
6 % EXTRETS -> N elements de L en forma de LLISTA
7 % CUA -> L - els N primers elements de L en forma de LLISTA
8 treuN(0, L, [], L).
9 treuN(N, L, EXTRETS, CUA):- K is N-1, K >= 0, append([A], B, L), treuN(
  K, B, NOUEX, CUA), append([A], NOUEX, EXTRETS), !.
```

Al final tenim una matriu cúbica representada en llistes aniuades on cada casella té una llista de variables que representen la possibilitat que hi hagi un valor en concret en aquella casella.

Execució

Fem un exemple concret i petit que no forma part de la resolució del sudoku de l'exemple.

```

1 ?- treuN(4, [1,2,3,4,5,6,7,8], EXTRETS, CUA).
2 EXTRETS = [1, 2, 3, 4],
3 CUA = [5, 6, 7, 8].
```

Inicialitzar

Aquest predicat tradueix els inputs representats en llista de c(F,C,V) en una CNF amb clàusules unitàries on cada clàusula representa el fet que es fixi un valor en concret en el tauler generat.

Predicat

```

1 %%%%%%%%%%%%%%%
2 % inicialitzar(N,LI,F)
3 % Donat el domini de les caselles (N), i la llista d'inicialitzacions (
  de forma [c[1,2,5],...]),
4 % -> el tercer parametre sera la CNF formada de clausules unitaries que
  forcen els valors corresponents
```

```

5 % a les caselles corresponents (als N-dominis corresponents)
6 inicialitzar(_,[],[]):-!.
7 inicialitzar(N,LI,CNF):- append([c(F,C,D)],Resta,LI),
8                             POS is N*N*(F-1) + (C-1)*N+D,
9                             inicialitzar(N,Resta,W),
10                            append([[POS]],W,CNF).

```

Execució

```

1 ?- inicialitzar(4, [c(1,1,3),c(2,2,1),c(4,4,3)], CNF).
2 CNF = [[3], [21], [63]].

```

Codifica sudoku

El que es proposa amb aquest predicat és generar les cnf que codifiquen que només hi ha un nombre possible per cada casella (kdominis), les cnf que codifiquen que tots els nombres de les caselles son diferents a cada una de les files, les cnf que codifiquen que tots els nombres de les caselles son diferents a cada una de les columnes i les cnf que codifiquen que tots els nombres de les caselles son diferents per cada un dels subquadrats del sudoku.

Els appends següents empalmen les diferents cnf per produir la CNF del paràmetre per tot el sudoku.

Predicat

```

1 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2 % codificaSudoku(N,Tauler,C0,CNF)
3 % Donat el domini de les caselles (N) d'un Sudoku, el seu tauler i la
4 % CNF que l'inicialitza (C0),
5 % -> el quart parametre sera la CNF que codifica el Sudoku.
6 codificaSudoku(N,Tauler,C0,CNF):- kdominis(Tauler,C1),
7                                   alldiffFiles(Tauler,C2),
8                                   alldiffColumnes(Tauler,C3),
9                                   alldiffQuadrats(Tauler,N,C4),
10                                  append(C1,C2,C12), append(C3,C4,C34)
                                   ,
                                   append(C0,C12,C012), append(C012,C34
                                   ,CNF).

```

Execució

```

1 ?- festauler(4, T), codificaSudoku(4, T, [[3],[21],[63]], CNF).

```

```

2  T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
        [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
        31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
        [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
        [61|...]]],
3  CNF = [[3], [21], [63], [1, 2, 3, 4], [-1, -2], [-1, -3], [-1, -4],
        [-2|...], [...|...]|...].

```

Crides internes

- K dominis
- All Diff Files
- All Diff Columnes
- All Diff Quadrats

K dominis

Reb el tauler i codifica el “exactament un” per a cada casella monta la codificació conforme només accepta un valor tal com es mostra en l'exemple d'execució següent.

Predicat

```

1  %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2  % kdominis(T,F)
3  % Donat un Tauler,
4  % -> el segon parametre es la CNF que codifica els exactamentUn per
    cada casella (K-domini)
5  kdominis([],[]).
6  kdominis([[_|T],F):-kdominis(T,F).
7  kdominis(T,F):- append([PF],CUA,T), append([PE],CUAFILA,PF),
    exactamentUn(PE,CNF1),
8      append([CUAFILA],CUA,RESTA),kdominis(RESTA,CNFFINAL),
    append(CNF1,CNFFINAL,F),!.

```

Execució

Podem observar que la resposta CNF és part de la resposta CNF del predicat codifica sudoku.

```

1  ?- festauler(4,T), kdominis(T,CNF).
2  T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
        [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
        31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
        [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
        [61|...]]],

```

```
3 CNF = [[1, 2, 3, 4], [-1, -2], [-1, -3], [-1, -4], [-2, -3], [-2, -4],
        [-3, -4], [5|...], [...|...]|...].
```

Crides internes

- Exactament Un

Exactament Un

Per cada una de les caselles és cridat aquest predicat que col·labora en la generació de la CNF per codificar el sudoku sencer. Valida que una casella només prengui un dels seus valors possibles.

Una CNF “exactament un” que codifica un únic nombre per la primera casella d’un sudoku de N=4 pren la forma que es veu en l’exemple d’execució d’aquest mateix apartat. Aquest exemple d’execució es realitza sobre el cas explicat al llarg de l’informe però només per una de les caselles del tauler.

Predicat

```
1 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2 % exactamentUn(L,CNF)
3 % Donat una llista de variables booleanes,
4 % -> el segon parametre sera la CNF que codifica que exactament una
    sigui certa.
5 exactamentUn([],[]).
6 exactamentUn([A],[[A]]).
7 exactamentUn(L,CNF):- negat(L, LNEG), montaParelles(LNEG,PARELLES),
    append([L],PARELLES, CNF).
```

Execució

Veiem l’execució només per la primera casella d’un sudoku n=4.

```
1 ?- exactamentUn([1,2,3,4],CNF).
2 CNF = [[1, 2, 3, 4], [-1, -2], [-1, -3], [-1, -4], [-2, -3], [-2, -4],
        [-3, -4]].
```

Crides internes

- Negat
- Monta Parelles

Negat

Les CNF's estan formades de llistes de llistes de k dominis on els k dominis son nombres enters. Cada nombre enter simbolitza un nombre donat una fila i una columna. Per tant codifica una variable, el fet de que la variable sigui certa o falça es representa amb un enter positiu o negatiu. La feina d'aquest predicat és la de capgirar el valor de les variables i passar-les de certes a falces o viceversa. I per fer-ho s'utilitza el 0 per restar-li el valor enter negatiu o positiu en funció de l'entrada.

Predicat

```
1 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2 % negat(A,ANEG)
3 % Donat una llista de literals enters torna una llista amb els literals
  enters negats.
4 negat([],[]).
5 negat([A],[ANEG]):- ANEG is 0-A,!.
6 negat([A|CUA], [ANEG|CUANEG]):- negat([A],[ANEG]), negat(CUA,CUANEG).
```

Execució

```
1 ?- negat([1,-2,-3,4],NEGAT).
2 NEGAT = [-1, 2, 3, -4].
```

Monta Parelles

La idea és que donada una llista amb n variables es cridi al predicat combina que genera una llista amb la combinació d'una de les variables amb la resta, fent així una llista de parelles de variables. Veure l'especificació del Combina.

La idea és que a mesura que es vagin combinant les variables no es tornint a combinar repetits i per tant s'exclou un cop combinada la variable i es torna a cridar el montaParelles amb la resta de variables que no poden haver-se combinat.

Predicat

```
1 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2 % montaParelles(L, PARELLES)
3 % PARELLES és el resultat de combinar tots els elements de L entre sí.
4 montaParelles([],[]).
5 montaParelles([A,B],[[A,B]]):-!.
6 montaParelles(L, PARELLES):- append([_],CUA,L),
7                               % [_] ja no ens importa per que ja s'ha combinat amb l'
                               element extret.
```

```

8      combina(L,PIVOTA), % [[1,2],[1,3],[1,4]]
9      montaParelles(CUA,ITSEG),% Montarà amb
      [2,3,4]
10     append(PIVOTA,ITSEG,PARELLES),!.

```

Execució

```

1  ?- montaParelles([1,2,3,4], PARELLES).
2  PARELLES = [[1, 2], [1, 3], [1, 4], [2, 3], [2, 4], [3, 4]].

```

Crides internes

- Combina

Combina

Donades una llista de variables, agafarà la primera i la combinarà amb la resta (només la primera). Ja es tornarà a cridar més endavant sense el primer element que ja s'ha combinat.

Predicat

```

1  %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2  % combina(L,COMBINAT)
3  % COMBINAT és la llista resultant de combinar el primer element de L
   amb la resta
4  % L -> és una llista d'elements a tractar
5  combina([],[]).
6  combina([A],[[A]]).
7  combina([A,B],[[A,B]]).
8  combina(L, COMBINAT):- append([A,B], CUA, L),
9                          append([A],CUA,PIVOTA),
10     combina(PIVOTA,ITSEG), append([[A,B]],ITSEG,
                                   COMBINAT),!.

```

Execució

```

1  ?- combina([1,2,3,4],COMBINAT).
2  COMBINAT = [[1, 2], [1, 3], [1, 4]].

```

All Diff Files

Controla que donades totes les files de la matriu, no hi hagi cap element repetit en cap de les files, complint així el primer dels tres propòsits d'un sudoku (No hi ha elements repetits a les files, no hi ha elements repetits a les columnes, no hi ha elements repetits als quadrats). CNF la llista de clàusules resultat que verifica que no hi ha cap repetit a les files.

Predicat

```

1  %%%%%%%%%%
2  % allDiffFiles(T,F)
3  % Donat un Tauler,
4  % -> el segon parametre es la CNF que codifica que no hi hagi repetits
    (allDiff) als K-dominis de cada fila
5  allDiffFiles([],[]).
6  allDiffFiles(T,F):- append([PF],CUA,T),
7                        allDiff(PF,CNF),
8                        allDiffFiles(CUA,CNFCUA),
9                        append(CNF,CNFCUA,F),!.
```

Execució

```

1  ?- festauler(4,T), allDiffFiles(T, CNF).
2  T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
    [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
    31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
    [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
    [61|...]]],
3  CNF = [[-1, -5], [-1, -9], [-1, -13], [-5, -9], [-5, -13], [-9, -13],
    [-2, -6], [-2|...], [...|...]|...].
```

Crides internes

- All Diff

All Diff

All diff reb una llista de llistes de variables que representen un subconjunt de caselles. Aquest subconjunt de caselles pot ser un a fila, una columna o bé un quadrat. El que codifica el alldiff és que totes les caselles que entren siguin diferents. Com que el format d'entrada és una llista i nosaltres volem combinar les diferents opcions que hi ha dins del llistat de variables el que fem és transposar aquest llistat per que el seu format passi d'aquí :


```
1  [[1,2,3,4],[5,6,7,8],[9,10,11,12],[13,14,15,16]]
```

A aquí:

```
1  [[1,5,9,13],[2,6,10,14],[3,7,11,15],[4,8,12,16]]
```

Realitzant aquesta transposició ja podem accedir directament a vincular els diferents elements cridant al montaparelles per cada una de les possibilitats forçant que per exemple només hi hagi un 1, o només hi hagi un 2 o només hi hagi un 3 ... fin a N.

Predicat

```
1  %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2  % allDiff(L,F)
3  % Donat una llista (L) de Kdominis,
4  % -> CNF codifica que no poden prendre el mateix valor.
5  % - Extraurem el primer literal de cada k-domini
6  % - muntem una llista amb tots els literals extrets
7  % - cridem a alldif amb totes les cues.
8  allDiff(L,CNF):-matTransposa(L,T), iAllDiff(T,CNF).
9
10
11 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
12 % INMERSIÓ (AHUUUUUHA)
13 % iAllDiff(L,CNF)
14 % Donada una llista de caselles L
15 % -> CNF serà la clausla que valida que totes les caselles son
    diferents
16 iAllDiff([],[]).
17 iAllDiff(L,CNF):-append([PCOL],MAT,L), negat(PCOL, PCOLNEG),
    montaParelles(PCOLNEG, PCOLNEGCMB),
18     iAllDiff(MAT, CNFPARCIAL), append(PCOLNEGCMB, CNFPARCIAL, CNF)
    ,!.
```

Execució

Codifiquem la primera fila del sudoku N=4.

```
1  ?- allDiff([[1,2,3,4],[5,6,7,8],[9,10,11,12],[13,14,15,16]], CNF).
2  CNF = [[-1, -5], [-1, -9], [-1, -13], [-5, -9], [-5, -13], [-9, -13],
    [-2, -6], [-2|...], [...|...]|...].
```

Crides internes

- Mat Transposa
- Negat
- Monta Parelles

Mat Transposa

Donada qualsevol matriu, fa la seva transposada i la posa a RES. Per fer-ho es realitza fila per fila, fent que la fila es divideix en una fila per cada element i unint el resultat de cridar recursivament a la transposada de la cua per després unir-la per files.

Predicat

```

1  %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2  % matTransposa(MAT,RES)
3  % Donat una matriu MAT.
4  % -> RES es la matriu transposada de MAT
5  matTransposa([],[]):-!.
6  matTransposa(MAT,RES):- append([PF],CUA,MAT),
7                          transposarFila(PF,PFT),
8                          matTransposa(CUA,REST),
9                          matUnio(PFT,REST,RES),!.
```

Execució

Fem un exemple de transposar la primera fila de una matriu de N=4.

```

1  ?- matTransposa([[1,2,3,4],[5,6,7,8],[9,10,11,12],[13,14,15,16]],
2  TRANSPOSADA).
3  TRANSPOSADA = [[1, 5, 9, 13], [2, 6, 10, 14], [3, 7, 11, 15], [4, 8,
4  12, 16]].
```

Un altre exemple seria transposar tot el tauler a nivell de Files i columnes, que tracta els dominins de cada casella no com una llista sinó com un element.

```

1  ?- festauler(4,T), matTransposa(T, TRANSPOSADA).
2  T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
3  [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
4  31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
5  [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
6  [61|...]]],
7  TRANSPOSADA = [[[1, 2, 3, 4], [17, 18, 19, 20], [33, 34, 35, 36], [49,
8  50, 51, 52]], [[5, 6, 7, 8], [21, 22, 23, 24], [37, 38, 39, 40],
9  [53, 54, 55|...]], [[9, 10, 11, 12], [25, 26, 27, 28], [41, 42,
```

```
43|...], [57, 58|...]], [[13, 14, 15, 16], [29, 30, 31|...], [45,
46|...], [61|...]]].
```

Crides internes

- Transposar Fila
- Mat Unió

Transposar Fila

Donada una fila es retorna una llista de files on cada element de la llista L és una sola llista dins de RES.

Predicat

```
1 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2 % transposarFila(L,RES)
3 % Donada una llista (fila) L
4 % -> RES es el resultat de dividir la fila en columnes i retorna com
    una matriu
5 transposarFila([],[]).
6 transposarFila(L,RES):- append([P],CUA,L),
7                          transposarFila(CUA,CUARES),
8                          append([[P]],CUARES,RES), !.
```

Execució

```
1 ?- transposarFila([[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13,
    14, 15, 16]], FILA_TRANSPOSADA).
2 FILA_TRANSPOSADA = [[1, 2, 3, 4], [5, 6, 7, 8]], [[9, 10, 11, 12]],
    [[13, 14, 15, 16]]].
```

Mat Unió

Donades dues matrius ESQ i DRE, el matunió genera la unió de les dues matriu a partir de unir-les fila a fila. És a dir la matriu $[[A],[C]]$ U $[[B],[D]] = [[A,B],[C,D]]$.

Aquesta unió és genèrica per qualsevol matriu ESQ i qualsevol Matriu DRE.

Predicat

```
1 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
```

```

2 % matUnio(ESQ,DRE,MAT)
3 % Donades 2 matrius ESQ i PDRE
4 % -> MAT es la unió per files de les 2 matrius
5 % -> EX: matUnio([[1]],[[2],[3]],[[1,2],[3]])
6 matUnio([],T,T).
7 matUnio(T,[],T).
8 matUnio(ESQ,DRE,MAT):-append([PESQ],CUAESQ,ESQ), append([PDRE],CUADRE,
    DRE), append(PESQ,PDRE,PF),
9                               matUnio(CUAESQ,CUADRE,MATRES),
10                              append([PF],MATRES,MAT),!.

```

Execució

Els següents exemplifiquen el comportament del mat unió. (És la unió de dues matrius de tota la vida unides per files).

```

1 ?- matUnio([[1],[2]],[[3]], RES).
2 RES = [[1, 3], [2]].
3
4 ?- matUnio([[1],[2]],[[[3]]], RES).
5 RES = [[1, [3]], [2]].
6
7 ?- matUnio([[1]],[[3],[4]], RES).
8 RES = [[1, 3], [4]].
9
10 ?- matUnio([[1],[3]],[],[4], RES).
11 RES = [[1], [3, 4]].

```

All Diff Columnnes

Simplement transposar la matriu que tenim, i cridar al all diff files. És un clar exemple d'abstracció.

Predicat

```

1 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2 % allDiffColumnnes(T,F)
3 % Donat un Tauler,
4 % -> el segon parametre es la CNF que codifica que no hi hagi repetits
    (allDiff) als K-dominis de cada columna
5 allDiffColumnnes(MAT,F):- matTransposa(MAT,T),
6                               allDiffFiles(T,F),!.

```

Execució

Veiem que ha transposat la matriu i que ha combinat els diferents parells de variables per que no hi hagi cap element repetit a les columnes.

```
1 ?- festauler(4,T), allDiffColumnes(T, CNF).
2 T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
      [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
      31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
      [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
      [61|...]]],
3 CNF = [[-1, -17], [-1, -33], [-1, -49], [-17, -33], [-17, -49], [-33,
      -49], [-2, -18], [-2|...], [...|...]|...].
```

Crides internes

- Mat Transposa
- All diff Files

All Diff Quadrats

Aquest predicta extreu un llistat de caselles per a cada possible quadrat del sudoku. Plantegem un sistema de quadrats basat en una fórmula matemàtica utilitzant les arrels de N, on N és el nombre del sudoku.

Aquests quadrats segueixen la lògica expressada en el següent esquema :

1	---	---	---	---	
2	0	0	1	1	
3	0	0	1	1	
4	2	2	3	3	
5	2	2	3	3	
6	---	---	---	---	

D'aquesta manera el primer quadrat estarà format per les caselles c(1,1), c(1,2), c(2,1), c(2,2).

El procediment que es segueix és el d'agafar fila per fila i verificar que les caselles estan dins del subquadrat que es demana. Per aquest motiu s'utilitza inmersió. Per poder generar una llista de valors enters de 0 fins a N-1 que permeti demanar per cada valor el subquadrat X del sudoku on $0 \leq X < N$.

Utilitzem el subset per generar el llistat de caselles que s'han de combinar per generar la CNF corresponent a cada quadrat.

Predicat

```

1 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2 % allDiffQuadrats(T,N,F)
3 % Donat un Tauler, i la mida del K-domini (que es tambe el nombre de
   quadrats que hi ha),
4 % -> el tercer parametre es la CNF que codifica que no hi hagi repetits
   (allDiff) als K-dominis de cada quadrat
5 allDiffQuadrats([],_,[]).
6 allDiffQuadrats(T,N,F):- iAllDiffQuadrats(T,N,0,F).
7
8 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
9 % iAllDiffQuadrats(T,N,M,F)
10 % Inmersió per allDiffQuadrats.
11 iAllDiffQuadrats(_,N,N,[]).
12 iAllDiffQuadrats(T,N,M,F):- X is M+1,
13                               subset(T,N,M,S),
14                               allDiff(S,CNF),
15                               iAllDiffQuadrats(T,N,X,CNFX),
16                               append(CNF,CNFX,F),!.

```

Execució

```

1 ?- N=4, festauler(N,T), allDiffQuadrats(T, N, CNF).
2 N = 4,
3 T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
   [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
   31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
   [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
   [61|...]]],
4 CNF = [[-1, -5], [-1, -17], [-1, -21], [-5, -17], [-5, -21], [-17,
   -21], [-2, -6], [-2|...], [...|...]|...].

```

Crides internes

- Subset
- All Diff

Subset

A partir de una matriu, extraurem el llistat de caselles S corresponent a el subquadrat Q demanat per paràmetre. Per fer-ho es situa un inici en funció del nombre del quadrat demanat seguint la fórmula:

$\text{fila} = (Q \bmod \text{SQRT}(N)) * \text{SQRT}(N)$ $\text{columna} = (Q / \text{SQRT}(N)) * \text{SQRT}(N)$

UN cop calculats els inicis passem tant el tauler com els inicis a un predicat extreu que acaba extreient tots els elements que estan compresos entre els inicis i els inicis + SQRT(N).

Predicat

```

1  %%%%%%%%%%%%%%
2  % subset(T,N,Q,S)
3  % Serveix per extreure Un subquadrat amb el valor SUBQUADRAT.
4  % T és tauler
5  % N és el valor del sudoku.
6  % S Serà el subset referent al subquadrat de numero subquadrat.
7  % -----
8  % | 0 | 0 | 1 | 1 |
9  % | 0 | 0 | 1 | 1 |
10 % | 2 | 2 | 3 | 3 |
11 % | 2 | 2 | 3 | 3 |
12 % -----
13 % SUBQUADRAT pot prendre valor de 1 - N on en la matriu expressada
    anteriorment cada valor
14 % refereix al subquadrat marcat amb els nombres de 1 a 4.
15 subset(T,N,SUBQUADRAT,S):- P is truncate(sqrt(N)),
16                               X is (SUBQUADRAT mod P)*P,
17                               Y is (SUBQUADRAT div P)*P,
18                               extreu(T,X,Y,N,S).
```

Execució

Veiem la crida de subset per cada un dels quadrats en un sudoku N=4.

```

1  ?- festauler(4,T), subset(T,4,0,SUBSET).
2  T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
    [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
    31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
    [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
    [61|...]]],
3  SUBSET = [[1, 2, 3, 4], [5, 6, 7, 8], [17, 18, 19, 20], [21, 22, 23,
    24]].
4
5  ?- festauler(4,T), subset(T,4,1,SUBSET).
6  T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
    [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
    31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
    [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
    [61|...]]],
```

```

7 SUBSET = [[9, 10, 11, 12], [13, 14, 15, 16], [25, 26, 27, 28], [29, 30,
  31, 32]].
8
9 ?- festauler(4,T), subset(T,4,2,SUBSET).
10 T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
  [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
  31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
  [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
  [61|...]]],
11 SUBSET = [[33, 34, 35, 36], [37, 38, 39, 40], [49, 50, 51, 52], [53,
  54, 55, 56]].
12
13 ?- festauler(4,T), subset(T,4,3,SUBSET).
14 T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
  [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
  31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
  [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
  [61|...]]],
15 SUBSET = [[41, 42, 43, 44], [45, 46, 47, 48], [57, 58, 59, 60], [61,
  62, 63, 64]].

```

Crides internes

- Extreu

Extreu

L'objectiu de l'extreu és que donats dos inicis referents a la fila i la columna extregui els elements compressos entre els inicis i els inicis + SQRT(N) que son els que delimiten el quadrat.

Es realitza una verificació de tots els elements del tauler (tallant quan s'ha s'ha extret tot el quadrat) per cada quadrat que es vol extreure. Comprovant si la fila i la columna avaluats en aquest moment estan compresos dins del rang estipulat entre inici i final per files i columnes.

Un cop hem trobat tots els elements d'una fila, la fila es suprimeix i es segueix amb la cua que son la resta de files de la matriu. Si No s'ha trobat l'element encara es van passant elements i files fins arribar a dins del rang. En cas que es trobi tots els elements de dins del quadrat i encara quedin files per mirar es talla.

Predicat

```

1 %%%%%%%%%%%%%%%
2 % extreu(T,XI,YI,N,S)

```



```

3 % Extreu el subquadrat que estigui entre XI - XI + SQRT(N) i YI - YI +
  SQRT(N) on X refereix a files i Y a columnes.
4 % T és Tauler
5 % XI és Fila inicial
6 % YI és columna inicial
7 % N és el valor del SUDOKU
8 % S serà el subquadrat de tamany SQRT(N) que està entre XI - XI + SQRT
  (N) i YI - YI + SQRT(N).
9 extreu(T,XI,YI,N,S):- ARR is truncate(sqrt(N)),
10                      XF is XI + ARR, YF is YI+ARR,
11                      iExtreu(T,XI,YI,XF,YF,0,0,S), !.
12
13 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
14 % iExtreu(T,XI,YI,XF,YF,X,Y,S)
15 % Inmersió d'extreu. Retorna el subquadrat entre XI i XF, YI i YF.
16 % T és Tauler
17 % XI és Fila inici, XF és fila Final, X és Fila actual
18 % YI és columna inici, YF és columna Final, Y és columna actual
19 % S és el subquadrat format per tots els elements entre XI i XY, YI i
  YF.
20
21 % ja ha trobat tot el quadrat
22 iExtreu(_,_,_,_,YF,_,YF,[]):-!.
23 % Encara no he arribat a la fila que cerco.
24 iExtreu(T,XI,YI,XF,YF,_,Y,S):- Y < YI, append([_],CUAF,T), YSEG is Y
  +1,
25                      iExtreu(CUAF,XI,YI,XF,YF,0,YSEG,S).
26 % me passat de columnes (X) puc saltar a la línia següent
27 iExtreu(T,XI,YI,XF,YF,XF,Y,S):- append([_],CUAF,T), YSEG is Y+1,
28                      iExtreu(CUAF,XI,YI,XF,YF,0,YSEG,S).
29 % No me passat ni de files ni de columnes i miro si estic dins del rang
  (extrec)
30 iExtreu(T,XI,YI,XF,YF,X,Y,S):- append([PF],CUAF,T), append([PC],
  CUAC, PF),
31                      dinsRang(XI,XF,YI,YF,X,Y),
32                      append([CUAC],CUAF,MAT), XSEG is X+1,
33                      iExtreu(MAT,XI,YI,XF,YF,XSEG,Y,Q),
34                      append([PC],Q,S), !.
35 % Encara no he trobat l'inici del rang i vaig passant elements
36 iExtreu(T,XI,YI,XF,YF,X,Y,S):- append([PF],CUAF,T),
37                      append([_], CUAC, PF),
38                      append([CUAC],CUAF,MAT), X<XF, XS is
  X+1,

```

39

`!Extreu(MAT,XI,YI,XF,YF,XS,Y,S).`**Execució**

```

1 ?- festauler(4, T), extreu(T,2,2,4,QUADRAT).
2 T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
      [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
      31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
      [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
      [61|...]]],
3 QUADRAT = [[41, 42, 43, 44], [45, 46, 47, 48], [57, 58, 59, 60], [61,
      62, 63, 64]].

```

Crides internes

- Dins Rang

Dins Rang

Únicament comprova que x i y estan dintre de Xi i XF i Yi i Yf.

Predicat

```

1 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2 % dinsRang(XI,XF,YI,YF,X,Y)
3 % Comprova que X i Y estan entre XI i XF, YI i YF els finals no
   inclosos.
4 % XI és Fila inici, XF és fila Final, X és Fila actual
5 % YI és columna inici, YF és columna Final, Y és columna actual
6 dinsRang(XI,XF,YI,YF,X,Y):- X>=XI, X<XF, Y>=YI, Y<YF.

```

Execució

```

1 ?- dinsRang(0,2,0,2,1,1).
2 true.
3
4 ?- dinsRang(0,2,0,2,3,3).
5 false.

```

Crides internes

Sat

Mira la satisfactibilitat de una CNF. utilitza l'algoritme de simplificació de variables eliminant clàusules de dins de la CNF a mesura que va decidint literals. Cada literal que fixa el valor per aquell literal com a part d'un model. Llavors elimina totes les clàusules on apareix aquest literal, i quan es troba ell mateix negat, el treu de la clàusula. Va aplicant aquest algoritme fins que es queda sense clàusules i per tant completa el model.

Si entre el les clàusules restants es troba la clàusula buida és que s'ha trobat un contraexemple i que per tant no té solució per a la interpretació donada fins el moment. Llavors el prolog genera el backtracking per veure si en alguna altre branca de l'arbre de cerca la CNF és satisfactible.

Predicat

```

1  %%%%%%%%%%%%%%
2  % sat(F,I,M)
3  % si F es satisfactible, M sera el model de F afegit a la interpretació
   I (a la primera crida I sera buida).
4  % Assumim invariant que no hi ha literals repetits a les clausules ni
   la clausula buida inicialment.
5  sat([],I,I):-      write('SAT!! , SUDOKU PISCINAS!!! '),nl,!.
6  sat(CNF,I,M):-
7      % Ha de triar un literal 'duna clausula unitaria, si no 'nhi ha cap
   , llavors un literal pendent qualsevol.
8      decideix(CNF,Lit),
9
10     % Simplifica la CNF amb el Lit triat (compte pq pot fallar, es a
   dir si troba la clausula buida fallara i fara backtraking).
11     simplif(Lit,CNF,CNFS),
12
13     % crida recursiva amb la CNF i la interpretacio actualitzada
14     append(I,[Lit],ISEG),
15     sat(CNFS,ISEG ,M).
```

Execució

En el següent exemple d'execució li demanarem al Sat tots els possibles models entrant “;”. No considerem que sigui una bona idea tallar aquest sat per que així podem aconseguir diferents models donada una CNF que representa el sudoku.

```

1  ?- taulerSudoku(4, [c(1,1,3),c(2,2,1),c(4,4,3)],T, C), codificaSudoku
   (4, T, C, CNF), sat(CNF, [], MODEL).
2  SAT!! , SUDOKU PISCINAS!!!
```

```

3  T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
      [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
      31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
      [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
      [61|...]]],
4  C = [[3], [21], [63]],
5  CNF = [[3], [21], [63], [1, 2, 3, 4], [-1, -2], [-1, -3], [-1, -4],
      [-2|...], [...|...|...],
6  MODEL = [3, 21, 63, -1, -2, -4, -22, -23, -24|...] ;
7  SAT!!, SUDOKU PISCINAS!!!
8  T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
      [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
      31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
      [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
      [61|...]]],
9  C = [[3], [21], [63]],
10 CNF = [[3], [21], [63], [1, 2, 3, 4], [-1, -2], [-1, -3], [-1, -4],
      [-2|...], [...|...|...],
11 MODEL = [3, 21, 63, -1, -2, -4, -22, -23, -24|...] ;
12 SAT!!, SUDOKU PISCINAS!!!
13 T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
      [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
      31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
      [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
      [61|...]]],
14 C = [[3], [21], [63]],
15 CNF = [[3], [21], [63], [1, 2, 3, 4], [-1, -2], [-1, -3], [-1, -4],
      [-2|...], [...|...|...],
16 MODEL = [3, 21, 63, -1, -2, -4, -22, -23, -24|...] ;
17 SAT!!, SUDOKU PISCINAS!!!
18 T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
      [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
      31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
      [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
      [61|...]]],
19 C = [[3], [21], [63]],
20 CNF = [[3], [21], [63], [1, 2, 3, 4], [-1, -2], [-1, -3], [-1, -4],
      [-2|...], [...|...|...],
21 MODEL = [3, 21, 63, -1, -2, -4, -22, -23, -24|...] ;
22 SAT!!, SUDOKU PISCINAS!!!
23 T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
      [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
      31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],

```

```

        [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
        [61|...]]],
24  C = [[3], [21], [63]],
25  CNF = [[3], [21], [63], [1, 2, 3, 4], [-1, -2], [-1, -3], [-1, -4],
        [-2|...], [...|...]|...],
26  MODEL = [3, 21, 63, -1, -2, -4, -22, -23, -24|...] ;
27  SAT!!, SUDOKU PISCINAS!!!
28  T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
        [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
        31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
        [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
        [61|...]]],
29  C = [[3], [21], [63]],
30  CNF = [[3], [21], [63], [1, 2, 3, 4], [-1, -2], [-1, -3], [-1, -4],
        [-2|...], [...|...]|...],
31  MODEL = [3, 21, 63, -1, -2, -4, -22, -23, -24|...].

```

Crides internes

- Decideix
- Simplif

Decideix

Si trobem una clàusula amb un sol literal, tirem pel dret amb aquest literal, altrament tirem pel dret amb el primer literal de la primera clàusula o el primer literal de la primera clàusula negat.

Predicat

```

1  %%%%%%%%%%%%%%
2  % decideix(F, Lit)
3  % Donat una CNF,
4  % -> el segon parametre sera un literal de CNF
5  % - si hi ha una clausula unitaria sera aquest literal, sino
6  % - un qualsevol o el seu negat.
7  % detectar clàusula unitària
8  decideix(L,A):- append(_, [X|_], L),
9  % append([A],[_],X),!.
10 % Retorna el primer element
11 decideix([PL|_], A) :- append([A],[_],PL).
12 % Retorna el negat del primer element
13 decideix([PL|_], B) :- append([A],[_],PL), B is 0-A.

```

Execució

```

1 | ?- decideix([[3],[21],[63],[1,2,3,4]], LITERAL).
2
3 LITERAL = 3
4
5 yes
6 | ?- decideix([[1,2,3,4]], LITERAL).
7
8 LITERAL = 1 ? ;
9 LITERAL = -1

```

Simplif

Va treien totes les clàusules on hi aparegui el literal que li passem i va treient els negats d'aquest literal de dins de les clàusules.

Predicat

```

1 %%%%%%%%%%%%%%
2 % simplif(Lit, CNF, CNFS)
3 % Donat un literal Lit i una CNF,
4 % -> el tercer parametre sera la CNF que ens han donat simplificada:
5 % - sense les clausules que tenen lit
6 % - treient -Lit de les clausules on hi es, si apareix la clausula
   buida fallara.
7 simplif(_, [], []). % QUAN TINGUEM LA LLISTA BUIDA RETORNEM LA LLISTA
   BUIDA.
8
9                                     % Comprovem si conté lit a primera
10 simplif(Lit, [PRIMERA|CUA], CNFS) :- member(Lit, PRIMERA),
11                                     % seguim amb la cua.
12                                     simplif(Lit, CUA, CNFS), !.
13                                     % neguem el literal i el fotem a x
14 simplif(Lit, [PRIMERA|CUA], CNFS) :- X is 0-Lit,
15                                     append(A, [X|XS], PRIMERA),
16                                     % Treure el literal negat a
17                                     primera serà res
18                                     append(A, XS, RES),
19                                     % crida recursiva
20                                     simplif(Lit, CUA, CUACNFS),
21                                     % Muntem la cadena final
22                                     append([RES], CUACNFS, CNFS), !.

```

```

21                                     % Crida recursiva quan no conté
                                     Lit
22 simplif(Lit, [PRIMERA|CUA], CNFS) :- simplif(Lit, CUA, CUACNFS),
23                                     % Concatenar resultats
24                                     append([PRIMERA], CUACNFS, CNFS),
                                     !.

```

Execució

```

1 ?- taulerSudoku(4, [c(1,1,3),c(2,2,1),c(4,4,3)],T, C), codificaSudoku
   (4, T, C, CNF), simplif(3, CNF, RESTA_CNF).
2 T = [[[1, 2, 3, 4], [5, 6, 7, 8], [9, 10, 11, 12], [13, 14, 15, 16]],
   [[17, 18, 19, 20], [21, 22, 23, 24], [25, 26, 27, 28], [29, 30,
   31|...]], [[33, 34, 35, 36], [37, 38, 39, 40], [41, 42, 43|...],
   [45, 46|...]], [[49, 50, 51, 52], [53, 54, 55|...], [57, 58|...],
   [61|...]]],
3 C = [[3], [21], [63]],
4 CNF = [[3], [21], [63], [1, 2, 3, 4], [-1, -2], [-1, -3], [-1, -4],
   [-2|...], [...|...]|...],
5 RESTA_CNF = [[21], [63], [-1, -2], [-1], [-1, -4], [-2], [-2, -4],
   [-4], [...|...]|...].

```

Mostra Sudoku

Mostra un sudoku ja sigui complert o incomplert a partir d'una entrada per teclat amb el format [c(F1,C1,V1)..c(Fn,Cn,Vn)].

Predicat

```

1 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2 % mostraSudoku(N, IN)
3 % Mostra un sudoku expressat en format llistat de c(F,C,V) per pantalla
   .
4 % N és el tamany del sudoku
5 % IN és el sudoku representat en llistat de c(F,C,V).
6 mostraSudoku(N, IN) :- pintaSeparador(N), nl, iMostraSudoku(N,1,1,IN).
7
8 iMostraSudoku(N, F, _, _):- F>N, !.
9 iMostraSudoku(N, F, C, IN):- C > N, write('|'), nl, pintaSeparador(N),
   nl,
10                                     FSEG is F+1, iMostraSudoku(N, FSEG, 1, IN)
   .

```

```

11 iMostraSudoku(N, F, C, IN):- pintaCasella(F,C,IN), CSEG is C+1,
12                               iMostraSudoku(N, F, CSEG, IN), !.
13 iMostraSudoku(N, F, C, IN):- CSEG is C+1, write('| '),
14                               iMostraSudoku(N, F, CSEG, IN).

```

Execució

```

1  ?- mostraSudoku(4, [c(1,1,3),c(2,2,1),c(4,4,3)]).
2  -----
3  |3|  |  |  |
4  -----
5  |  |1|  |  |
6  -----
7  |  |  |  |  |
8  -----
9  |  |  | 3|
10 -----
11 true.

```

Crides internes

- Pinta Separador
- Pinta casella

Pinta Separador

Pinta $(N+2)+(N-1)$ guions com una sola línia.

Predicat

```

1  %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2  % pintaSeparador(N)
3  % Pinta un separador generat amb '-' amb tamany N+2+N-1 o N*2+1 carà
   cters
4  pintaSeparador(N):- NF is (N+2)+(N-1), iPintaSeparador(1, NF).
5
6  %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
7  % iPintaSeparador(N, NF)
8  % Crida immersiva del pintaSeparador.
9  iPintaSeparador(N, NF):- N>NF, !.
10 iPintaSeparador(N, NF):- write('-'), NSEG is N+1, iPintaSeparador(NSEG,
   NF).

```


Execució

```

1  ?- pintaSeparador(4).
2  -----
3  true.

```

Pinta Casella

Cerca dins de la llista in si hi ha una casella que coincideixi amb files i columnes a partir del l'append i si hi és el pinta. Si no hi és retorna no.

Predicat

```

1  %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
2  % pintaCasella(F,C,IN, FCUA)
3  % Pinta el valor de una casella amb el separador al davant si
   coincideixen F = IN[F] i C = IN[C].
4  % F és el valor de la fila, C és el valor de la Columna
5  % IN son les caselles possibles a pintar per la casella [F,C]
6  % FCUA seran les caselles restants si es compleix que es pot pintar.
7  pintaCasella(F,C,IN) :- append(_, [c(F,C,V)|_], IN), write('|'), write(V
   ), !.

```

Execució

```

1  ?- pintaCasella(1,1,[c(1,1,3),c(2,2,1),c(4,4,3)]).
2  |3
3  true.
4
5
6  ?- pintaCasella(1,2,[c(1,1,3),c(2,2,1),c(4,4,3)]).
7  false.

```

Exemples i jocs de proves.

Hi ha una sèrie de fitxers que comencen per la paraula sudoku i que son de tipus txt on hi ha diferents jocs de proves.

SUDOKUS AMB SOLUCIÓ

- **sudoku1.txt** -> Té solució
- **sudoku2.txt** -> Té solució

sudoku1

```

1  | ?- resol(9,[c(1,2,5),c(1,5,8),c(1,8,3),c(2,2,2),c(2,3,3),c(2,4,9),c
    (2,6,5),c(2,7,7),c(3,1,9),c(3,3,6),c(3,4,7),c(4,2,4),c(4,3,2),c
    (4,5,7),c(4,7,8),c(4,8,9),c(5,1,6),c(5,9,5),c(6,2,9),c(6,3,1),c
    (6,5,5),c(6,7,4),c(6,8,6),c(7,6,1),c(7,7,9),c(7,9,3),c(8,3,4),c
    (8,4,3),c(8,6,6),c(8,7,5),c(8,8,7),c(9,2,3),c(9,5,4),c(9,8,1)]) .

2
3  NEM A RESOLDRE EL SUDOKU :
4  .....
5  -----
6  | |5| | |8| | |3| |
7  -----
8  | |2|3|9| |5|7| | |
9  -----
10 |9| |6|7| | | | | |
11 -----
12 | |4|2| |7| |8|9| |
13 -----
14 |6| | | | | | |5|
15 -----
16 | |9|1| |5| |4|6| |
17 -----
18 | | | | | |1|9| |3|
19 -----
20 | | |4|3| |6|5|7| |
21 -----
22 | |3| | |4| | |1| |
23 -----
24 SAT!!, SUDOKU PISCINAS!!!
25 .....
26 -----
27 |4|5|7|1|8|2|6|3|9|
28 -----
29 |1|2|3|9|6|5|7|8|4|
30 -----
31 |9|8|6|7|3|4|1|5|2|

```

```

32  -----
33  |5|4|2|6|7|3|8|9|1|
34  -----
35  |6|7|8|4|1|9|3|2|5|
36  -----
37  |3|9|1|2|5|8|4|6|7|
38  -----
39  |7|6|5|8|2|1|9|4|3|
40  -----
41  |2|1|4|3|9|6|5|7|8|
42  -----
43  |8|3|9|5|4|7|2|1|6|
44  -----

```

sudoku2

```

1  | ?- resol(9, [c(1,4,8), c(1,8,4), c(1,9,7), c(2,4,2), c(2,5,3), c
    (2,9,5), c(3,4,7), c(3,8,8), c(4,2,8), c(4,3,7), c(4,6,3), c(5,2,5),
    c(5,5,4), c(5,6,8), c(5,9,2), c(6,3,3), c(6,4,5), c(6,5,2), c(6,7,8)
    ,c(6,8,9), c(7,1,7), c(7,4,3), c(8,1,6), c(8,2,2), c(8,6,5), c
    (8,7,9) ,c(8,8,7), c(8,9,3), c(9,2,4), c(9,8,2), c(9,9,6)]).
2
3  NEM A RESOLDRE EL SUDOKU :
4  .....
5  -----
6  | | | 8 | | | 4 | 7 |
7  -----
8  | | | 2 | 3 | | | 5 |
9  -----
10 | | | 7 | | | 8 | |
11 -----
12 | 8 | 7 | | 3 | | | |
13 -----
14 | 5 | | 4 | 8 | | 2 |
15 -----
16 | | 3 | 5 | 2 | 8 | 9 | |
17 -----
18 | 7 | | 3 | | | | |
19 -----
20 | 6 | 2 | | | 5 | 9 | 7 | 3 |
21 -----

```

```

22 | |4| | | | |2|6|
23 -----
24 SAT!!, SUDOKU PISCINAS!!!
25 .....
26 -----
27 |9|1|2|8|5|6|3|4|7|
28 -----
29 |8|7|4|2|3|9|1|6|5|
30 -----
31 |5|3|6|7|1|4|2|8|9|
32 -----
33 |2|8|7|1|9|3|6|5|4|
34 -----
35 |1|5|9|6|4|8|7|3|2|
36 -----
37 |4|6|3|5|2|7|8|9|1|
38 -----
39 |7|9|5|3|6|2|4|1|8|
40 -----
41 |6|2|1|4|8|5|9|7|3|
42 -----
43 |3|4|8|9|7|1|5|2|6|
44 -----
45
46 true ? ;
47
48 (3984 ms) no

```

SUDOKUS SENSE SOLUCIÓ

- **sudokuNoQ.txt** -> No té solució per que té elements repetits en un quadrat.
- **sudokuNoF.txt** -> No té solució per que té elements repetits en una fila.
- **sudokuNoC.txt** -> No té solució per que té elements repetits en una columna.

sudokuNoQ

```

1 | ?- resol(9,[c(1,2,5),c(1,5,8),c(1,8,3),c(2,2,2),c(2,3,3),c(2,4,9),c
    (2,6,5),c(2,7,7),c(3,1,9),c(3,3,6),c(3,4,7),c(4,2,4),c(4,3,2),c
    (4,5,7),c(4,7,8),c(4,8,9),c(5,1,6),c(5,9,5),c(6,2,9),c(6,3,1),c
    (6,5,5),c(6,7,4),c(6,8,6),c(7,6,1),c(7,7,9),c(7,9,3),c(8,3,4),c

```

```

      (8,4,3),c(8,6,6),c(8,7,5),c(8,8,7),c(9,2,3),c(9,5,4),c(9,8,1), c
      (9,9,9)]).
2
3  NEM A RESOLDRE EL SUDOKU :
4  .....
5  -----
6  | |5| | |8| | |3| |
7  -----
8  | |2|3|9| |5|7| | |
9  -----
10 |9| |6|7| | | | | |
11 -----
12 | |4|2| |7| |8|9| |
13 -----
14 |6| | | | | | |5|
15 -----
16 | |9|1| |5| |4|6| |
17 -----
18 | | | | | |1|9| |3|
19 -----
20 | | |4|3| |6|5|7| |
21 -----
22 | |3| | |4| | |1|9|
23 -----
24
25 (2094 ms) no

```

sudokuNoF

```

1  | ?- resol(9,[c(1,2,5),c(1,5,8),c(1,8,3),c(2,2,2),c(2,3,3),c(2,4,9),c
      (2,6,5),c(2,7,7),c(3,1,9),c(3,3,6),c(3,4,7),c(4,2,4),c(4,3,2),c
      (4,5,7),c(4,7,8),c(4,8,9),c(5,1,6),c(5,9,5),c(6,2,9),c(6,3,1),c
      (6,5,5),c(6,7,4),c(6,8,6),c(7,6,1),c(7,7,9),c(7,9,3),c(8,3,4),c
      (8,4,3),c(8,6,6),c(8,7,5),c(8,8,7),c(9,2,3),c(9,5,4),c(9,8,1), c
      (9,9,4)]).
2
3  NEM A RESOLDRE EL SUDOKU :
4  .....
5  -----
6  | |5| | |8| | |3| |
7  -----

```

```

 8 | | 2|3|9| | 5|7| | |
 9 -----
10 | 9| | 6|7| | | | |
11 -----
12 | | 4|2| | 7| | 8|9| |
13 -----
14 | 6| | | | | | | 5|
15 -----
16 | | 9|1| | 5| | 4|6| |
17 -----
18 | | | | | | 1|9| | 3|
19 -----
20 | | | 4|3| | 6|5|7| |
21 -----
22 | | 3| | | 4| | | 1|4|
23 -----
24
25 (2109 ms) no

```

sudokuNoC

```

1 | ?- resol(9,[c(1,2,5),c(1,5,8),c(1,8,3),c(2,2,2),c(2,3,3),c(2,4,9),c
    (2,6,5),c(2,7,7),c(3,1,9),c(3,3,6),c(3,4,7),c(4,2,4),c(4,3,2),c
    (4,5,7),c(4,7,8),c(4,8,9),c(5,1,6),c(5,9,5),c(6,2,9),c(6,3,1),c
    (6,5,5),c(6,7,4),c(6,8,6),c(7,6,1),c(7,7,9),c(7,9,3),c(8,3,4),c
    (8,4,3),c(8,6,6),c(8,7,5),c(8,8,7),c(9,2,3),c(9,5,5),c(9,8,1)]) .
2
3 NEM A RESOLDRE EL SUDOKU :
4 .....
5 -----
6 | | 5| | | 8| | | 3| |
7 -----
8 | | 2|3|9| | 5|7| | |
9 -----
10 | 9| | 6|7| | | | |
11 -----
12 | | 4|2| | 7| | 8|9| |
13 -----
14 | 6| | | | | | | 5|
15 -----
16 | | 9|1| | 5| | 4|6| |

```

```

17  -----
18  | | | | | 1|9| |3|
19  -----
20  | | |4|3| |6|5|7| |
21  -----
22  | |3| | |5| | |1| |
23  -----
24
25  (2141 ms) no

```

Comentaris sobre la solució i aspectes positius.

Abstracció

Com és d'esperar, en les nostres pràctiques mirem d'abstreure les funcionalitats per reutilitzar-les en diferents parts del programa. Per exemple podem observar les anstraccions dels predicats `montaParelles` o del `matTransposa` i fins i tot del `allDiff` entre altres.

Talls.

S'ha procurat que el prolog faci la feina que ha de fer i cap altre. Per demostrar-ho s'ha realitzat un exemple d'execució de cada predicat per que e pugui veure la solució que es proposa i juntament amb la definició del predicat es pugui interpretar que no hi ha més opcions possibles.

S'ha de descartar el tall en el sat, ja que el sat interessa que torni models si el cridem independentment de la resta de predicats.

L'us de l'append

L'append és el més utilitzat per el tractament de llistes en aquesta pràctica.

Cerques

L'ús de l'append per fer una cerca en un llistat per exemple de caselles (Cerca si hi ha la casella que coincideix en files i columnes a dins de IN i la pinta):

```

1  pintaCasella(F,C,IN) :- append(_, [c(F,C,V) | _], IN), write('|'), write(V), !.

```

Detectar clàusula unitària

Podem usar-la per detectar si hi ha una llista amb un sol element dins d'una llista.

```
1 decideix(L,A):- append(_, [X|_], L), append([A], [], X), !.
```

Construir i desconstruir Llistes

És molt usat en general per poder desengranar les llistes a plaer. També per construir-les.

```
1 combina(L, COMBINAT):- append([A,B], CUA, L),  
2                          append([A], CUA, PIVOTA),  
3                          combina(PIVOTA, ITSEG),  
4                          append([[A,B]], ITSEG, COMBINAT), !.
```

Conclusió

La pràctica de prolog ha sigut molt clara i fàcil d'interpretar un cop estudiat l'esquelet de la pràctica. Reconeixem que sense l'esquelet ens hi hauríem fet bastant de mal. Ens ha agradat la manera que té el prolog de treballar, com evalúa i genera l'arbre de cerca i com realitza el backtracking ell solet. És quelcom curiós i amb aplicació directe a problemes que son molt interessants.